STAT 525 FALL 2018

Chapter 19
Two-Factor Studies with Equal Sample Sizes

Professor Min Zhang

Two-Factor Studies

- Now have two factors (A and B)
- Suppose each factor has two levels
  - Also known as a 2^2 factorial
  - Could analyze as one factor with 4 levels
    * Trt 1: A high, B high
    * Trt 2: A high, B low
    * Trt 3: A low, B high
    * Trt 4: A low, B low
  - Use contrasts to test for A or B effect

$$A \text{ effect } = \frac{\text{Trt1} + \text{Trt2}}{2} - \frac{\text{Trt3} + \text{Trt4}}{2}$$

Example

An experiment is conducted to study the effect of hormones injected into test rats. There are two distinct hormones (A,B) each with two distinct levels. For purposes here, we will consider this to be four different treatments labeled {A,a,B,b}. Each treatment is applied to six rats with the response being the amount of glycogen (in mg) in the liver.

Treatment	Responses					
A	106	101	120	86	132	97
a	51	98	85	50	111	72
B	103	84	100	83	110	91
b	50	66	61	72	85	60

Three contrasts are of interest. They are:

Comparison	A	a	B	b
Hormone A vs Hormone B	1	1	-1	-1
Low level vs High level	1	-1	1	-1
Equivalence of level effect	1	-1	-1	1

Can we reanalyze the experiment in such a way that these sum of squares are already separated?

The Data in Two-Factor Studies

- Y is the response variable
- Factor A has levels $i = 1, 2, ..., a$
- Factor B has levels $j = 1, 2, ..., b$
- Y_{ijk} is the k^{th} observation from cell (i, j)
- Often presented as a table

	Level of B			
Level of A	1	2	$\cdots$	b
1	XXXXXX	XXXXXX	$\cdots$	XXXXXX
2	XXXXXX	XXXXXX	$\cdots$	XXXXXX
$\vdots$	$\vdots$	$\vdots$	$\vdots$	$\vdots$
a	XXXXXX	XXXXXX	$\cdots$	XXXXXX

- Chapter 19 assumes $n_{ij} = n$
- Chapter 20 assumes $n_{ij} = 1$
- Chapter 23 allows n_{ij} to vary

Example (Page 833)

- Castle Bakery supplies wrapped Italian bread to a large number of supermarkets
- Bakery interested in the set up of their store display
  - Height of display shelf (top, middle, bottom)
  - Width of display shelf (regular, wide)
- Twelve stores equal in sales volume were selected
- Randomly assigned equally to each of 6 combinations
- Y is the sales of the bread
  - $i = 1, 2, 3$ and $j = 1, 2$
  - $n_{ij} = n = 2$

```
data a1;
    infile 'u:\.www\datasets525\CH19TA07.txt';
    input sales height width;
    if height eq 1 and width eq 1 then hw='1_BR';
    if height eq 1 and width eq 2 then hw='2_BW';
    if height eq 2 and width eq 1 then hw='3_MR';
    if height eq 2 and width eq 2 then hw='4_MW';
    if height eq 3 and width eq 1 then hw='5_TR';
    if height eq 3 and width eq 2 then hw='6_TW';

/* Scatterplot */
symbol1 v=circle i=none;
proc gplot data=a1;
    plot sales*hw/frame;
run; quit;
```



The Model

- Same basic assumptions as regression
  - All observations assumed independent
  - All observations normally distributed with
    - a mean that may depend on the levels of factors A and B
    - constant variance
- Often presented in terms of **cell means** or **factor effects**

The Cell Means Model

- Expressed numerically

$$Y_{ijk} = \mu_{ij} + \varepsilon_{ijk}$$

  - μ_{ij} is the theoretical mean or expected value of all observations in cell (i, j)
  - The ε_{ijk} are iid $N(0, \sigma^2)$ which implies the Y_{ijk} are independent $N(\mu_{ij}, \sigma^2)$

- Parameters
  - $\{\mu_{ij}\}$, $i = 1, 2, .., a$, $j = 1, 2, ..., b$
  - σ^2

Estimates / Inference

- Estimate μ_{ij} by the sample mean of the observations in cell (i, j)

$$\widehat{\mu}_{ij} = \overline{Y}_{ij.}$$

- For each cell (i, j), also estimate of the variance

$$s_{ij}^2 = \sum_k (Y_{ijk} - \overline{Y}_{ij.})^2/(n-1)$$

- These s_{ij}^2 are combined to estimate σ^2

ANOVA Table : $n_{ij} = n$

- Similar ANOVA table construction
- Plug in $\overline{Y}_{ij.}$ as fitted value

Source of Variation	df	SS
Model	$ab-1$	$n\sum_i\sum_j(\overline{Y}_{ij.}-\overline{Y}_{...})^2$
Error	$ab(n-1)$	$\sum_i\sum_j\sum_k(Y_{ijk}-\overline{Y}_{ij.})^2$
Total	$abn-1$	$\sum_i\sum_j\sum_k(Y_{ijk}-\overline{Y}_{...})^2$

$$\overline{Y}_{...} = \sum_i\sum_j\sum_k Y_{ijk}/abn$$

$$\overline{Y}_{ij.} = \sum_k Y_{ijk}/n$$

- Can further break down into Factor A, Factor B and interaction effects using contrasts

Factor Effects Model

- Statistical model is

$$Y_{ijk} = \mu + \alpha_i + \beta_j + (\alpha\beta)_{ij} + \varepsilon_{ijk}, \quad \begin{cases} i = 1, 2, \ldots, a \\ j = 1, 2, \ldots, b \\ k = 1, 2, \ldots, n \end{cases}$$

- Over-parameterized model
  - Must include $a + b + 1$ model constraints
  - Conceptual Approach

$$\textstyle\sum_i \alpha_i = 0, \quad \sum_j \beta_j = 0, \quad \sum_i (\alpha\beta)_{ij} = 0, \quad \sum_j (\alpha\beta)_{ij} = 0$$

    * μ - grand mean
    * α_i - ith level effect of factor A (ignores B)
    * β_j - jth level effect of factor B (ignores A)
    * $(\alpha\beta)_{ij}$ - interaction effect of combination ij

Relation to the Cell Means Model

- Breaks down cell means

$$\mu = \sum_i \sum_j \mu_{ij}/(ab)$$

$$\alpha_i = \mu_{i.} - \mu \text{ with } \mu_{i.} = \sum_j \mu_{ij}/b$$

$$\beta_j = \mu_{.j} - \mu \text{ with } \mu_{.j} = \sum_i \mu_{ij}/a$$

$$(\alpha\beta)_{ij} = \mu_{ij} - (\mu + \alpha_i + \beta_j)$$

- Interaction effect is the difference between the cell means model and the additive (or main effects) model. Explains variation not explained by main effects.

- SAS puts a different set of $a + b + 1$ constraints

$$\alpha_a = 0, \quad \beta_b = 0, \quad (\alpha\beta)_{aj} = 0, \quad (\alpha\beta)_{ib} = 0$$

Factor Effects Estimates

- The $a + b + 1$ conceptual-approach constraints result in

$$\widehat{\mu} = \overline{Y}_{...}$$

$$\widehat{\alpha}_i = \overline{Y}_{i..} - \overline{Y}_{...}$$

$$\widehat{\beta}_j = \overline{Y}_{.j.} - \overline{Y}_{...}$$

$$(\widehat{\alpha\beta})_{ij} = \overline{Y}_{ij.} - \overline{Y}_{i..} - \overline{Y}_{.j.} + \overline{Y}_{...}$$

  - What about the SAS-approach constraints?

- The predicted value and residual are

$$\widehat{Y}_{ijk} = \overline{Y}_{ij.}$$

$$e_{ijk} = Y_{ijk} - \overline{Y}_{ij.}$$

Questions in Example (Page 833)

- Does the height of the display affect sales?
  - If yes, will need to do pairwise comparisons
- Does the width of the display affect sales?
  - If yes, will need to do pairwise comparisons
- Does the effect of height depend on the width?
  - If yes, we have an interaction
- Does the effect of width depend on the height?
  - If yes, we have an interaction

Partitioning the Sum of Squares

$$Y_{ijk} - \overline{Y}_{...} = (\overline{Y}_{i..} - \overline{Y}_{...}) + (\overline{Y}_{.j.} - \overline{Y}_{...}) + (\overline{Y}_{ij.} - \overline{Y}_{i..} - \overline{Y}_{.j.} + \overline{Y}_{...}) + (Y_{ijk} - \overline{Y}_{ij.})$$

- Consider $\sum_i \sum_j \sum_k (Y_{ijk} - \overline{Y}_{...})^2$
- Right hand side simplifies to

$$\begin{aligned} & bn\sum_i (\overline{Y}_{i..} - \overline{Y}_{...})^2 \\ + \quad & an\sum_j (\overline{Y}_{.j.} - \overline{Y}_{...})^2 \\ + \quad & n\sum_i \sum_j (\overline{Y}_{ij.} - \overline{Y}_{i..} - \overline{Y}_{.j.} + \overline{Y}_{...})^2 \\ + \quad & \sum_i \sum_j \sum_k (Y_{ijk} - \overline{Y}_{ij.})^2 \end{aligned}$$

- Can be written as

$$\text{SSTO} = \text{SSA} + \text{SSB} + \text{SSAB} + \text{SSE}$$

- Degrees of freedom also broken down
- Under normality, all SS/σ^2 independent
- ANOVA Table

Source of Variation	Sum of Squares	Degrees of Freedom	Mean Square
Factor A	SSA	$a-1$	MSA
Factor B	SSB	$b-1$	MSB
Interaction	SSAB	$(a-1)(b-1)$	MSAB
Error	SSE	$ab(n-1)$	MSE
Total	SSTO	$abn-1$	

Hypothesis Testing

- Can show: Fixed Case

  E(MSE)=σ^2

  E(MSA) $= \sigma^2 + bn\sum \alpha_i^2/(a-1)$

  E(MSB) $= \sigma^2 + an\sum \beta_j^2/(b-1)$

  E(MSAB) $= \sigma^2 + n\sum (\alpha\beta)_{ij}^2/(a-1)(b-1)$

- Use F-test to test for A, B, and AB effects

$$F^* = \frac{\text{SSA}/(a-1)}{\text{SSE}/(ab(n-1))}$$

$$F^* = \frac{\text{SSB}/(b-1)}{\text{SSE}/(ab(n-1))}$$

$$F^* = \frac{\text{SSAB}/(a-1)(b-1)}{\text{SSE}/(ab(n-1))}$$

Example (Page 833)

```
proc glm data=a1;
    class height width;
    model sales=height width height*width;
    means height width height*width;
run; quit;
```

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	5	1580.000000	316.000000	30.58	0.0003
Error	6	62.000000	10.333333		
Corrected Total	11	1642.000000			

R-Square	Coeff Var	Root MSE	sales Mean
0.962241	6.303040	3.214550	51.00000

Source	DF	Type I SS	Mean Square	F Value	Pr > F
height	2	1544.000000	772.000000	74.71	<.0001
width	1	12.000000	12.000000	1.16	0.3226
height*width	2	24.000000	12.000000	1.16	0.3747

Source	DF	Type III SS	Mean Square	F Value	Pr > F
height	2	1544.000000	772.000000	74.71	<.0001
width	1	12.000000	12.000000	1.16	0.3226
height*width	2	24.000000	12.000000	1.16	0.3747

Level of height	N	sales Mean	sales Std Dev
1	4	44.0000000	3.16227766
2	4	67.0000000	3.74165739
3	4	42.0000000	2.94392029

Level of width	N	sales Mean	sales Std Dev
1	6	50.0000000	12.0664825
2	6	52.0000000	13.4313067

Level of height	Level of width	N	sales Mean	sales Std Dev
1	1	2	45.0000000	2.82842712
1	2	2	43.0000000	4.24264069
2	1	2	65.0000000	4.24264069
2	2	2	69.0000000	2.82842712
3	1	2	40.0000000	1.41421356
3	2	2	44.0000000	2.82842712

Interaction Plot

```
proc means data=a1;
    var sales; by height width;
    output out=a2 mean=avsales;

symbol1 v=square i=join c=black; symbol2 v=diamond i=join c=black;
proc gplot data=a2;
    plot avsales*height=width/frame;
run; quit;
```



Results

- There appears to be no interaction between height and width (P=0.37) $\rightarrow$ The effect of width (or height) is the same regardless of height (or width). Because of this, we can focus on the main effects (averages out the other effect).

- The main effect for width is not statistically significant (P=0.32) $\rightarrow$ Width does not affect sales of bread

- The main effect for height is statistically significant ($P <$ 0.00001). From the scatterplot and interaction plot, it appears the middle location is better than the top and bottom.

- Pairwise testing (adjusting for multiple comparisons) can be performed.

Pooling Sums of Squares in Two-Factor ANOVA

- Some argue that an insignificant interaction should be dropped from the model (i.e., pooled with error

$$\text{SSE}^* = \text{SSE} + \text{SSAB}$$

$$\text{df}_{\text{E}}^* = ab(n-1) + (a-1)(b-1)$$

- Increases DF but could inflate $\widehat{\sigma}^2$

- Affects both the level of significance and power of testing main effects, in ways not yet fully understood.

- Should be considered when

  - The test statistic $MSAB/MSE$ falls substantially below the action limit of the decision rule (say $MSAB/MSE < 2$ for $\alpha = 0.05$, or reporting large p-value like > 0.25);

  - The degrees of freedom associated with MSE are small, perhaps 5 or less.

```
/*------------ WITHOUT POOLING ------------*/
proc glm data=a1;
    class height width;
    model sales=height|width;
    means height / tukey lines;
run; quit;
```

```
Tukey's Studentized Range (HSD) Test for sales

Error Degrees of Freedom                          6
Error Mean Square                          10.33333
Critical Value of Studentized Range         4.33902
Minimum Significant Difference                6.974

           Mean      N    height
A        67.000      4    2

B        44.000      4    1
B
B        42.000      4    3
```

```
/*------------ POOLING ------------*/
proc glm data=a1;
    class height width;
    model sales=height width;
    means height / tukey lines;
run; quit;
```

```
Tukey's Studentized Range (HSD) Test for sales

Error Degrees of Freedom                          8
Error Mean Square                             10.75
Critical Value of Studentized Range         4.04101
Minimum Significant Difference               6.6247

           Mean      N    height
A        67.000      4    2

B        44.000      4    1
B
B        42.000      4    3
```

Interaction in Two-Factor ANOVA

General Plan for Two-Factor ANOVA

- Construct scatterplot / interaction plot
- Run full model
- Check assumptions
  - Residual plots
  - Histogram / QQplot
  - Ordered residuals plot
- Check significance of interaction

If Interactions Are Not Significant

- Determine whether pooling is beneficial
  - If yes, rerun analysis without interaction
- Check significance of main effects
  - If factor insignificant, determine whether pooling is beneficial
    * If yes, rerun analysis as one-way ANOVA
  - If statistically significant factor has more than two levels, use multiple comparison procedure to assess differences
    * Contrasts and linear combinations can also be used

Typical Interactions

- Opposite behavior (no Factor 2 effect?)



Plot #1

- Not similar increase (still Factor 2 effect?)



Plot #2

If Interactions Are Significant But Not Important

- Plots and a careful examination of the cell means may indicate that the interaction is not very important even though it is statistically significant.
  - The interaction effects may be much smaller in magnitude than the main effects;
  - The interaction effects may only be apparent in a small number of treatments as in Plot #2.
  - The subject area specialist (researcher) needs to be consulted in deciding whether an interaction is important or unimportant
- Use the marginal means for each significant main effect to describe the important results for the main effects, but carefully interpret the marginal means as averages over the levels of the other factor and not a main effect.
- Keep the interaction in the model.
- KNNL also discuss ways that transformations can sometimes eliminate interactions.

If Interactions Are Significant and Important

The interaction effect is so large and/or pervasive that main effects cannot be interpreted on their own. Options include the following:

- Can take the approach of one-way ANOVA with ab levels to compare factor level means. Use linear combinations to compare various means (e.g., levels of factor A for each level of factor B).
- Use the interaction plots for discussion purposes on interactions between factors.
- Performs one-way ANOVA on one factor for a fixed level of the other factor.

Using `ESTIMATE` Statement

- Must formulate in terms of factor effects model
- In SAS: order of factors determined by order in `CLASS` statement not `MODEL` statement
- Example from Castle Bread Company
  - $H_0 : \mu_{2.} = \mu_{1.} + \mu_{3.}$
  - Rewriting in factor effect terms

$$\begin{aligned}
\mu_{2.} &= (\mu_{21} + \mu_{22})/2 \\
&= (\mu + \alpha_2 + \beta_1 + (\alpha\beta)_{21})/2 + \\
&\quad (\mu + \alpha_2 + \beta_2 + (\alpha\beta)_{22})/2 \\
&= \mu + \alpha_2 + (\beta_1 + \beta_2 + (\alpha\beta)_{21} + (\alpha\beta)_{22})/2 \\
\mu_{1.} + \mu_{3.} &= (\mu_{11} + \mu_{12})/2 + (\mu_{31} + \mu_{32})/2 \\
&= 2\mu + \alpha_1 + \alpha_3 + \beta_1 + \beta_2 + ((\alpha\beta)_{11} + (\alpha\beta)_{12} + (\alpha\beta)_{31} + (\alpha\beta)_{32})/2
\end{aligned}$$

  - Rewriting $\mu_{2.} - (\mu_{1.} + \mu_{3.})$
    * $-\mu$
    * $-\alpha_1 + \alpha_2 - \alpha_3$
    * $-\beta_1/2 - \beta_2/2$
    * $-(\alpha\beta)_{11}/2 - (\alpha\beta)_{12}/2 + (\alpha\beta)_{21}/2 + (\alpha\beta)_{22}/2 - (\alpha\beta)_{31}/2 - (\alpha\beta)_{32}/2$

Using `SLICE` Option in `LSMEANS`

- `SLICE` option performs one-way ANOVA on one factor for a fixed level of the other factor
  - Test for the effect of A within each level of B (as A*B is significant)

    ```
    lsmeans A*B / slice=B;
    ```
- Can also express that as contrast statement
- The following example considers two contrasts
  - $H_0 : \mu_{2.} = (\mu_{1.} + \mu_{3.})/2$
  - $H_0 : \mu_{11} = \mu_{21} = \mu_{31}$
- See if you can come up with the same contrast statements

Example (Page 833)

```
options nocenter ls=75;
proc glm data=a1;
    class height width;
    model sales=height width height*width;
    estimate 'h2-(h1+h3)'
      intercept -1
      height -1 1 -1
      width -.5 -.5
      height*width -.5 -.5 .5 .5 -.5 -.5;
    contrast 'h2 vs (h1+h3)/2'
      height -.5 1 -.5
      height*width -.25 -.25 .5 .5 -.25 -.25;
    estimate 'h2-(h1+h3)/2'
      height -.5 1 -.5
      height*width -.25 -.25 .5 .5 -.25 -.25;
    contrast 'mu11=mu21=mu31'
      height 1 -1 0 height*width 1 0 -1 0 0 0,
      height 0 1 -1 height*width 0 0 1 0 -1 0;
    means height*width;
run; quit;
```

Contrast	DF	Contrast SS	Mean Square	F Value	Pr > F
h2 vs (h1+h3)/2	1	1536.000000	1536.000000	148.65	<.0001
mu11=mu21=mu31	2	700.000000	350.000000	33.87	0.0005

Parameter	Estimate	Standard Error	t Value	Pr > |t|
h2-(h1+h3)	-19.0000000	2.78388218	-6.83	0.0005
h2-(h1+h3)/2	24.0000000	1.96850197	12.19	<.0001

Level of height	Level of width	N	------------sales------------ Mean	Std Dev
1	1	2	45.0000000	2.82842712
1	2	2	43.0000000	4.24264069
2	1	2	65.0000000	4.24264069
2	2	2	69.0000000	2.82842712
3	1	2	40.0000000	1.41421356
3	2	2	44.0000000	2.82842712

```
/* Use SLICE Option */
proc glm data=a1;
    class height width;
    model sales=height width height*width;
    lsmeans height*width / slice=width;
run; quit;
```

height*width Effect Sliced by width for sales

width	DF	Sum of Squares	Mean Square	F Value	Pr > F
1	2	700.000000	350.000000	33.87	0.0005
2	2	868.000000	434.000000	42.00	0.0003

Chapter Review

- Data
- Model
- Strategies for Analysis
  - when interaction not present
  - when interaction present